
PROPOSED REVISION: 

1. Update locale concepts

2. Update SPECIAL-NAMES samples in concepts

JUSTIFICATION:

This proposal brings the concepts in line with reference 2.

COMPATIBILITY:

This proposal is compatible with ANSI X3.23-1985 as amended by ANSI X3.23a-1989 and ANSI X3.23b-1993.

DISCUSSION:

Initial Investigation

See reference 2.

Meeting 220

PROPOSED CHANGES TO BASE DOCUMENT:

1. Page 733, C.13.2.2 Using a specific collating sequence, sample code, change ALPHABET clause as follows:

“ALPHABET ASCII-Sort FOR ALPHANUMERIC IS STANDARD-1

ALPHABET USC-Sort FOR NATIONAL IS STANDARD-1”

2. Page 734, C.13.2.4 Selecting a collating sequence for SORT or MERGE, sample code, change ALPHABET clause as follows:

“ALPHABET Universal-order FOR NATIONAL IS STANDARD-3“
3. Pages 736 – 742, C.14 Culturally-specific, culturally-adaptable, and multilingual applications, replace by the following:

“C.14 Culturally-specific, culturally-adaptable, and multilingual applications

Culturally-specific applications are designed for the needs of one specific language or culture.

Culturally-adaptable applications are designed and coded once for users of many diverse languages and cultures, and are tailored at runtime to behave in ways suitable for a given language or culture.

Multilingual applications are designed and coded once to handle more than one language or culture in any given execution.

C.14.1 Culturally-specific applications

New features in this draft International Standard that make it easier to develop applications for one specific language or culture are:

— multiple-character currency signs and mixed-case currency signs;

— a class test for characters in a particular alphabet;

— support for selecting and using a specific locale.

To define a multiple-character currency sign, the CURRENCY SIGN clause in the SPECIAL-NAMES paragraph needs to be coded; for example:

SPECIAL-NAMES.

CURRENCY-SIGN IS "EUR " with PICTURE SYMBOL "u". *> note the space after EUR

Then the letter "u" can be used in picture character strings; for example: PIC u99.99 — which defines an item of length 9. For the value 10.00, the edited data item would contain EUR 10.00.

A culturally-specific class test can be used by first defining an alphabet-name associated with a code-set or with a locale and then referencing that alphabet-name in a class test. The following code fragment illustrates a class test for the characters defined in category LC_CTYPE in a locale named "Japan-123" assumed to be provided by the implementor:

ENVIRONMENT DIVISION.

SPECIAL-NAMES.

LOCALE Japanese IS Japan-123 
*> Give the locale mnemonic-name "Japanese"

*> for use in a SET statement

ALPHABET Japanese-letters FOR NATIONAL IS LOCALE Japanese. 

*> Associate alphabet-name

*> Japanese-letters with a locale

...

DATA DIVISION.

...

01 the-input-stream USAGE NATIONAL PIC N(100).

...

PROCEDURE DIVISION.

...

IF the-input-stream IS Japanese-letters THEN ...

*> the alphabet-name is specified in the class test

This class test is true if the only characters in the-input-stream are the characters defined in locale category LC_CTYPE in the current locale.

The following code fragment illustrates a class test using the coded character set associated with STANDARD-1 to test data of usage DISPLAY:

SPECIAL-NAMES.

    ALPHABET ok-data FOR ALPHANUMERIC IS STANDARD-1.

...

DATA DIVISION.

01 the-input-stream USAGE DISPLAY PIC N(80).

...

PROCEDURE DIVISION.

...

    IF the-input-stream IS ok-data THEN ...

...

This class test is true if all characters in the-input-stream are characters specified in the coded character set STANDARD-1.

C.14.2 Culturally-adaptable applications

The nature of a culturally-adaptable application is that it is written once and there is no need to recompile the source code in order for the application to be used on the same type of processor in different cultures. COBOL supports cultural adaptability for the following cultural elements:

— monetary formatting

— number formatting

— collating sequences of file indexes, sort/merge, and comparisons

— case classification of letters

— date and time formatting

To make an application culturally-adaptable, the details of these cultural elements are specified outside of COBOL where they can be easily selected at runtime. The details are specified in a "locale", which is a set of specifications built by a utility in accordance with an international standard -- either ISO/IEC 9945-2 or ISO/IEC 14652. The implementor may provide such a utility, it may come with the operating environment, or it may have to be purchased. The implementor may also provide some pre-built locales. Then, in each operating environment that supports locales, there is a system locale defined for the local culture and, typically, the capability for each application to have a user locale and even a variety of locales that can be selected during execution of the application.

This draft International Standard requires that at least one locale be provided by COBOL implementors in cases where the operating environment does not support locales. The user will need to know which locales are available to the application.

In designing an application for cultural adaptability, there is a lot more to consider than just the cultural elements supported by COBOL. For example, it may be necessary to choose field sizes that accommodate a variety of monetary values. It will help to isolate message text in tables or files that can be easily substituted. It is necessary to think about the elements that can vary across the cultures for which you are designing, and about how the details of those elements can be determined at runtime rather than at compile time.

C.14.2.1 Locale selection

At startup, a user-default locale is available to the run unit and will be used for locale-based processing for all locale categories in every COBOL runtime module in that run unit -- unless a specific locale is indicated for a given data element, function, or statement -- until a SET statement is executed to switch to another locale or a locale is specified in a LOCALIZE clause for a specific runtime module.

A SET statement can be used to switch locales for a single locale category or for all categories at once. For example, if the SYSTEM-DEFAULT locale is to be just for monetary formatting, a SET statement can be coded as follows:

SET LOCALE LC_MONETARY TO SYSTEM-DEFAULT

Other examples of locale selection are given below.

C.14.2.1.1 Switching locales in a COBOL runtime module

If a specific locale or multiple specific locales are to be used in an application, instead of the system default or user default locales, the locale needs to be identified in the SPECIAL-NAMES paragraph and a locale-name be assigned for use in COBOL procedural code. The details of how to identify a locale are specified by the COBOL implementor. A library of named locales might be provided, or an implementor could accept a data set location in a literal -- these are examples; see the user documentation for your COBOL compiler to find out what your implementor has provided.

The following code fragment illustrates the code for identifying and switching locales, assuming the implementor has provided locales named French-1 and Swiss-2:

PROGRAM-ID. IndexParts.

ENVIRONMENT DIVISION.

OBJECT-COMPUTER.

    PROGRAM COLLATING SEQUENCE FOR NATIONAL IS Locale-Sort.

*>  This causes the national collating sequence to be determined at runtime

*>  by the current locale -- the user-default unless a SET LOCALE statement has

*>  switched to another locale.

...

SPECIAL-NAMES.

*>  Name some locales for use later in SET LOCALE statements:

    LOCALE French IS "D:\LocaleLib\French-1" 
*> library example

    LOCALE Swiss IS Swiss-2 


*> another example, a named locale

...

*>  Associate the alphabet-name "Locale-Sort" with the locale facility:

    ALPHABET Locale-Sort FOR NATIONAL IS LOCALE.

...

DATA DIVISION.

...

01  default-locale-pointer USAGE POINTER. 
*> for saving the default locale

...

PROCEDURE DIVISION.

*>  The user-default locale is in effect initially. If you will need it again, save 

*>  a pointer identifying the user-default locale:

    SET default-locale-pointer TO LOCALE USER-DEFAULT.

...

*>  Now, switch to the locale you want to use.

*>  If you don't want to switch locales for all locale

*>  categories, specify a SET statement for each category you want to switch:

    SET LOCALE LC_COLLATE TO Swiss. *> switch program collating sequence

    SET LOCALE LC_MONETARY TO French *> switch monetary formatting

...

*>  Suppose now you have completed processing with the two locales and you want to 

*>  use the user default for a while:

    SET LOCALE LC_COLLATE TO USER-DEFAULT

    SET LOCALE LC_MONETARY TO USER-DEFAULT

...

*>  Now, suppose you want to switch the user default to one of your locales:

    SET LOCALE USER-DEFAULT TO Swiss

...

*>  Now, having finished your work with the Swiss locale, restore the previous

*>  USER-DEFAULT locale:

    SET LOCALE USER-DEFAULT TO default-locale-pointer.

...

It's a good idea to always save a pointer to the current locale before switching to a new one, unless it is known that no further processing will need that locale.

C.14.2.1.2 Switching locales outside of COBOL

The implementor may, but is not required to, support inter-operation of COBOL with other programming languages; one example is the C programming language. When support is provided, it is up to the implementor whether a locale switch for the user-default locale in an activated non-COBOL runtime module is capable of being recognized on return to COBOL. If supported, recognition in COBOL is not automatic; execution of a SET statement specifying the user-default locale is required in COBOL; for example, to access all categories of the new locale:

SET LOCALE LC_ALL TO USER-DEFAULT.

C.14.2.2 Locale-based monetary and numeric formatting

Locale-based monetary or numeric formatting is done by specifying the LOCALE phrase in the PICTURE clause in the data description entry of a data item.

To have an item edited entirely in accordance with the locale specification, regardless of its picture specification, a LOCALE phrase must be coded in the picture character string. When using the LOCALE phrase, the picture character string is not an indication of the field size needed to hold the edited item. The programmer must design for the largest size needed and code a SIZE phrase in the PICTURE clause.

For example:

01 US-amount PICTURE +$9.9 LOCALE SIZE IS 10.

results in a signed numeric-edited field 10 characters long that will be edited entirely in accordance with locale category LC_MONETARY. The currency sign and its placement, the sign convention, the grouping separator and placement, and the decimal separator and placement will be determined from the locale that is current at the time of editing. The locale category LC_MONETARY is used for all numeric formatting, both for monetary and non-monetary formatting.

If a specific locale is to be used for editing, a locale-name needs to be assigned to it in the SPECIAL-NAMES paragraph, for example, my-locale in the following:

SPECIAL-NAMES.

    LOCALE my-locale is USA-1.

*>  assuming the implementor has given you a way to define or reference a locale

*>  named USA-1

...

DATA DIVISION.

...

01  US-amount PICTURE +$9.9 LOCALE my-locale SIZE IS 10.

Multiple locales may be used; for example, if a report had US currency in one column and Italian currency in another, one could define:

LOCALE my-US-locale is USA-1.

LOCALE my-Italian-locale is Italian-1.*> assuming locales named USA-1 and Italian-1

...

DATA DIVISION.

01  US.

...

05  US-amount PICTURE +$9.9 LOCALE my-US-locale SIZE IS 8.

...

01  Italy.

...

05  Italian-amount PICTURE +$9.9 LOCALE my-Italian-locale SIZE IS 18.

...

C.14.2.3 Locale-based collating sequences

To use a collating sequence specified by a locale throughout the entire compilation unit, a PROGRAM COLLATING SEQUENCE clause must be coded specifying an alphabet-name that is associated with the locale facility. The following code fragment illustrates the selection of a locale for a national collating sequence and the selection of a standard alphanumeric collating sequence:

OBJECT-COMPUTER.

    PROGRAM COLLATING SEQUENCE FOR ALPHANUMERIC IS ASCII-Sort

    FOR NATIONAL IS UCS-Sort.

...

SPECIAL-NAMES.

    ALPHABET ASCII-Sort FOR ALPHANUMERIC IS STANDARD-1

    ALPHABET UCS-Sort FOR NATIONAL IS LOCALE.

This causes the runtime national collating sequence to be the collating sequence specified by locale category LC_COLLATE in the locale current during execution. The alphanumeric collating sequence is the collating sequence associated with STANDARD-1.

The switching of locales used for collating sequences is illustrated inC.14.2.1.1, Switching locales in a COBOL runtime module.

C.14.2.4 Locale-based case classification of letters

It is possible to have the case classification of letters determined by locale category LC_CTYPE, which defines character classification, case conversion, and other character attributes. To do this, one specifies LC_CTYPE in the LOCALIZE clause of the OPTIONS paragraph, for example:

OPTIONS.

    LOCALIZE LC_CTYPE.

This results in runtime use of the current locale for determining

· whether characters are ALPHABETIC, ALPHABETIC-UPPER, or ALPHABETIC-LOWER in a class test;

· whether the content of a data item is consistent with its picture string when checked with a VALIDATE statement;

· the case of characters for conversion using the UPPER-CASE intrinsic function or the LOWER-CASE intrinsic function.

Locale-based case conversion with the UPPER-CASE function or the LOWER-CASE function can result in the return of more or fewer characters than the argument, although the use of these functions without a locale always returns a string that is the same length as the argument.

C.14.2.5 Date and time formatting

Locale-based date and time formatting are provided by the intrinsic functions LOCALE-DATE and LOCALE-TIME, respectively.

To convert a given date field to a locale-based format in accordance with a specific locale, use the LOCALE-DATE function and specify the locale's mnemonic-name as the second argument; for example:

MOVE LOCALE-DATE (some-date-field, my-italian-locale) TO a-date-field

· assuming there is a repository entry like that illustrated inC.11, Boolean support and bit manipulation, making it unnecessary to code the word "FUNCTION" before "LOCALE-DATE".

To obtain the current date in a locale-based format in accordance with the current locale, use the LOCALE-DATE function without specifying a locale as the second argument; for example:

MOVE LOCALE-DATE ( CURRENT-DATE (1:8) ) TO a-date-field.

Similarly, one can obtain locale-based time using the LOCALE-TIME function.

C.14.3 Multilingual applications

Multilingual applications are more easily developed if your implementor supports UCS-2 or UCS-4, defined in ISO/ IEC 10646-1, as a native coded character set. With these coded character sets and support for standardized cultural ordering, defined in ISO/IEC 14651, applications can be designed that handle most languages because the characters of most languages and a culturally-meaningful collating sequence are available with little coding effort.

For example, if the implementor supports UCS-2 as a native coded character set, the data items can be defined as usage NATIONAL and an alphabet with any desired name, specifying STANDARD-3 for the collating sequence:

OBJECT-COMPUTER.

    PROGRAM COLLATING SEQUENCE FOR NATIONAL IS universal-order.

SPECIAL-NAMES.

    ALPHABET universal-order FOR NATIONAL IS STANDARD-3.

At the time this draft International Standard is being developed, devices that support UCS-2 and ordering support complying with ISO/IEC 14651 are not yet widely available, but their use is expected to increase over time because of the need for applications of international scope. The implementor's documentation needs to be checked before using these processor-dependent features.

Use of these international standards alone may not be sufficient for developing multi-lingual applications. If the application requires monetary or number formatting, date and time formatting, or case classification of letters, use of the locale-based features described inC.14.2, Culturally-adaptable applications should be considered.”
RESOLVED ISSUES:

1. None

OPEN ISSUES:

1. None

